


情報処理概論

工学部 物質科学工学科
応用化学コース
機能物質化学クラス

第10回

2005年 6月30日





サブルーチン

- 関数と同様,プログラムの一部を別に定義し,再利用を容易にする.
- 関数との違い: 引数の値を変更できる.
 - 複数の値を返す場合に利用.
(関数は一つの値(計算結果)しか返せない)

サブルーチンを利用した例(1/3)



```
program sample3
  implicit none
  integer :: number1, number2
  integer, dimension(:), allocatable :: english, math
  real(8), external :: average
  intrinsic dble

  open(10, file="eng.dat")
  read(10, *) number1
  allocate(english(number1))
  call input_data(10, number1, english)
  close(10)

  write(*, '(A20, F6.2)') "Ave. of English = ", &
    average(number1, english)

  open(11, file="math.dat")
  read(11, *) number2
  allocate(math(number2))
  call input_data(11, number2, math)
  close(11)
```

サブルーチンを利用した例 (2/3)



```
write(*, '(A20, F6.2)') "Ave. of Math = ", &  
                        average(number2, math)  
  
stop  
end program
```

```
subroutine input_data(file, n, a)  
  implicit none  
  integer, intent(IN) :: file  
  integer, intent(IN) :: n  
  integer, dimension(n), intent(OUT) :: a  
  integer :: i  
  
  do i = 1, n  
    read(file, *) a(i)  
  end do  
end subroutine
```

```
function average(n, a)  
  implicit none  
  integer, intent(IN) :: n  
  integer, dimension(n), intent(IN) :: a
```

サブルーチンを利用した例(3/3)



```
integer :: i, total
real(8) :: average

total = 0
do i = 1, n
    total = total + a(i)
end do

average = dble(total) / dble(n)

end function
```



サブルーチンの定義

- program ~ end program の外側で定義
 - 前でも後でも可
- サブルーチンには型が無いので宣言不要

```
subroutine サブルーチン名(引数)
  implicit none
  引数の型宣言
  サブルーチン内で用いる変数,関数の型宣言
```

... 処理 ...

```
end subroutine
```

```
subroutine input_data(file, n, a)
  implicit none
  integer, intent(IN) :: file
  integer, intent(IN) :: n
  integer, dimension(n), intent(OUT) :: a
  integer :: i

  do i = 1, n
    read(file, *) a(i)
  end do
end subroutine
```



サブルーチンの引数

- 呼び出し側とサブルーチン側で同じ位置に指示された変数や値が互いに対応

```
call input_data(10, number1, english)
```

```
subroutine input_data(file, n, a)
  implicit none
  integer, intent(IN) :: file
  integer, intent(IN) :: n
  integer, dimension(n), intent(OUT) :: a
```

- サブルーチン内で引数が書き換えられると対応する変数の値も変更
- サブルーチン内での扱われ方を宣言時に指定
 - intent(IN) 値を参照するだけ
 - intent(OUT) 書き換えだけ,もしくは書き換え後参照
 - intent(INOUT) 値を参照後に書き換え




サブルーチンの利用

- 宣言は不要

- call で呼び出し

```
call input_data(10, number1, english)
```

- 数式や write文の中では利用できない
- 注意：
intent(OUT), intent(INOUT)の引数には
変数を渡す
 - 数値や数式は不可(intent(IN)の引数なら可)
 - サブルーチン内で値が書き換えられるため



サブルーチンと関数

- サブルーチン: ”小人さん”のイメージ
 - 指定された一定の仕事を行う
 - 引数として渡された変数の値を変更できる
 - call で呼び出し(数式での利用不可)
 - 利用前の宣言不要
- 関数: 数学関数の延長
 - 与えられた引数に応じてひとつの値を算出する
 - 引数として渡した変数の値は変更不可
 - 数式内でも利用可能
 - 利用前に宣言が必要



メインプログラムとサブプログラム

- メインプログラム
 - 最初に実行される部分プログラム
 - 一つのプログラム中に一つだけ
- サブプログラム
 - 関数又はサブルーチン
 - メインプログラムやサブプログラムから呼び出される部分プログラム